

Using SAS® Studio Tasks to Plot with ODS Graphics











Philip R Holland, Holland Numerics Limited

ABSTRACT


SAS® Studio includes Tasks that can be used to generate SAS programs to process SAS data sets. The Graph Tasks generate SAS programs that use ODS Graphics to produce a range of plots and charts. However, SAS Enterprise Guide® 7.1 and SAS Add-In for Microsoft Office 7.1 can also make use of these SAS Studio Tasks to generate graphs with ODS Graphics, even though their built-in Tasks use SAS/GRAPH®. This paper describes these SAS Studio Graph Tasks.

INTRODUCTION

There have been papers already written about generating ODS Graphics code with the ODS Graphics Designer, or with PROC SGPLOT and PROC SGSCATTER, but this paper will compare the output from the most commonly used SAS code for producing plots and charts using SAS/GRAPH and ODS Graphics generated using a task included in SAS Studio. Note, however, that only the ODS Graphics programs which can be replaced with only a SAS Studio task or two have been included. The SAS Studio tasks used in this paper come from the following list included in SAS Studio 3.4 under Tasks > Graph, although more and more are being introduced as the product matures:

-  Bar Chart
-  Bar-Line Chart
-  Box Plot
-  Bubble Plot
-  Histogram
-  Line Chart
-  Pie Chart
-  Scatter Plot
-  Series Plot
-  Simple HBar

For users of SAS Enterprise Guide, who have access to a SAS Studio installation, these tasks are available there too, provided the following option for SAS Studio settings is used. The following screenshot shows the SAS Studio 3.4 Single User has been installed locally. If a suitable SAS Studio installation is available on a server, then the Enterprise Edition can be selected instead:




SCATTER PLOTS

The first and simplest of the commonly used plots is the scatter plot. Basically a lot of data points scattered over the graph area.


SAS/GRAPH FROM SAS 9.2

```
PROC GPLOT DATA = sashelp.class;  
  SYMBOL V = CIRCLE I = NONE;  
  PLOT weight * height = sex;  
RUN;
```



ODS GRAPHICS FROM THE SCATTER PLOT TASK

It should be noted that, because the input data is unsorted, and the first record contains a record where sex='M', the legend starts with sex='M'.



X variable: Height
Y variable: Weight
Group variable: Sex

```
/*--SGPLOT proc statement--*/  
proc sgplot data=SASHELP.CLASS;  
  /*--Scatter plot settings--*/  
  scatter x=Height y=Weight /  
    group=Sex transparency=0.0  
    name='Scatter';  
  /*--X Axis--*/  
  xaxis grid;  
  /*--Y Axis--*/  
  yaxis grid;  
run;
```


LINE PLOTS

The second and probably the most frequently used of the common plots is the line plot. SAS/GRAPH provides a vast range of options as to how the individual points are used to create the line, but the sample code is just joining each point to the next, which is why the points have been sorted by the values on the x-axis.

SAS/GRAPH FROM SAS 9.2


```
PROC SORT DATA = sashelp.class
    OUT = class;
    BY sex height;
RUN;

PROC GPLOT DATA = class;
    SYMBOL V = CIRCLE I = JOIN;
    PLOT weight * height = sex;
RUN;
```



ODS GRAPHICS FROM THE SORT DATA AND SERIES PLOT TASKS

It should be noted that, now the input data has been sorted, and the first record contains a record where sex='F', so the legend order will match that in the SAS/GRAPH plot. There are also fewer ticks on the 2 axes.



X variable: Height
Y variable: Weight
Group variable: Sex

```
/* From Sort Data task */  
proc sort data=SASHELP.CLASS  
    out=CLASS_SORTDS equals;  
    by Sex Height;  
run;  
  
/*--SGPLOT proc statement--*/  
proc sgplot  
    data=WORK.CLASS_SORTDS;  
    /*--Scatter plot settings--*/  
    series x=Height y=Weight /  
        group=Sex markers  
        transparency=0.0  
        name='Series';  
    /*--X Axis--*/  
    xaxis grid;  
    /*--Y Axis--*/  
    yaxis grid;  
run;
```


BOX PLOTS

The box plot can be used to display the simple statistics, including the quartiles and outliers.

SAS/GRAPH FROM SAS 9.2


The default SAS/GRAPH plot certainly requires a number of features to be customized, such as the box width, to make it useful for the viewer.

```
PROC SORT DATA = sashelp.class  
    OUT = class;  
    BY age;  
RUN;  
  
PROC GPLOT DATA = class;  
    SYMBOL I = BOX00T;  
    PLOT height * age;  
RUN;
```



ODS GRAPHICS FROM THE SORT DATA AND BOX PLOT TASKS

Using the default settings for ODS Graphics produces a box plot that is recognizably similar to that produced using the default settings in SAS/GRAPH, but where the viewer can see much more clearly the information in the graph.



Analysis variable: Height

Category variable: Age

Direction: Vertical

```
/* From Sort Data task */
proc sort data=SASHELP.CLASS
      out=CLASS_SORTDS equals;
  by Sex Height;
run;

/*--SGPLOT proc statement--*/
proc sgplot data=WORK.CLASS_SORTDS;
  /*--Box Plot settings--*/
  vbox Height / category=Age
      fillattrs=(color=CXCAD5E5)
      name='Box';
  /*--Category Axis--*/
  xaxis fitpolicy=splitrotate;
  /*--Response Axis--*/
  yaxis grid;
run;
```

VERTICAL BAR CHARTS


This is the first group of bar charts created by PROC GCHART in SAS/GRAPH. Each group will include a simple bar chart, a stacked bar chart (known as “sub-grouped” in SAS/GRAPH) and a clustered bar chart (known as “grouped” in SAS/GRAPH). All of the graphs in this paper are drawn in square graph areas.

SAS/GRAPH FROM SAS 9.2

The simple and stacked bar charts drawn by SAS/GRAPH appear unusually narrow, probably because the default bar widths are fixed, rather than adapting to the graph area.

```
PROC SORT DATA = sashelp.class
      OUT = class;
  BY sex age;
RUN;

PROC GCHART DATA = class;
  VBAR age / DISCRETE;
RUN;
```




```

PROC SORT DATA = sashelp.class
    OUT = class;
    BY sex age;
RUN;

PROC GCHART DATA = class;
    VBAR age / SUBGROUP = sex
    DISCRETE;
RUN;

```




```

PROC SORT DATA = sashelp.class
    OUT = class;
    BY sex age;
RUN;


PROC GCHART DATA = class;
    VBAR sex / GROUP = age
    PATTERNID = MIDPOINT;
RUN;

```



ODS GRAPHICS FROM THE BAR CHART TASK


Note that the identification of the bars in the stacked and cluster bar charts are solely with the legend.



Category variable: Age

Direction: Vertical

```
/*--SGPLOT proc statement--*/  
proc sgplot data=SASHELP.CLASS;  
  /*--Bar chart settings--*/  
  vbar Age / name='Bar';  
  /*--Response Axis--*/  
  yaxis grid;  
run;
```




Category variable: Age

Group variable: Sex

Direction: Vertical

Group layout: Stack

```
/*--SGPLOT proc statement--*/  
proc sgplot data=SASHELP.CLASS;  
  /*--Bar chart settings--*/  
  vbar Age / group=Sex  
             groupdisplay=Stack  
             name='Bar';  
  /*--Response Axis--*/  
  yaxis grid;  
run;
```

```

/*--SGPLOT proc statement--*/
proc sgplot data=SASHELP.CLASS;
  /*--Bar chart settings--*/
  vbar Age / group=Sex
            groupdisplay=Cluster
            name='Bar';
  /*--Response Axis--*/
  yaxis grid;
run;

```

Category variable: Age

Group variable: Sex

Direction: Vertical

Group layout: Cluster

HORIZONTAL BAR CHARTS

The default horizontal bar chart in SAS/GRAPH includes statistics at the right-hand side of the chart. This feature creates the biggest problem when attempting to replicate this chart exactly in ODS Graphics. However, most frequently this graph is created without any statistics.

SAS/GRAPH FROM SAS 9.2


Note that the default horizontal bar charts, unlike the vertical bar charts, make full use of the available graph area.

```

PROC SORT DATA = sashelp.class
  OUT = class;
  BY sex age;
RUN;


PROC GCHART DATA = class;
  HBAR age / DISCRETE;
RUN;

```



ODS GRAPHICS FROM THE BAR CHART TASK

If you just want to create a horizontal bar chart in the form of a “rotated” vertical bar chart, then this is certainly possible using SAS Studio Bar Chart task. Stacked and cluster versions of the horizontal bar chart can be recreated by just changing the Direction option used in the vertical bar charts to Horizontal.



Category variable: Age

Direction: Horizontal

```
/*--SGPLOT proc statement--*/  
proc sgplot data=SASHELP.CLASS;  
  /*--Bar chart settings--*/  
  hbar Age / name='Bar';  
  /*--Response Axis--*/  
  xaxis grid;  
run;
```


As an alternative the SAS Studio Simple HBar task can generate the same ODS Graphics code as above, but without the need to specify the Direction.

2D PIE CHARTS

2D pie charts are simple to create using SAS/GRAPH, and there are many options to manipulate the way the segments and labels are presented. The sample code used has been deliberately kept very simple.


SAS/GRAPH FROM SAS 9.2

```
PROC SORT DATA = sashelp.class  
  OUT = class;  
  BY age;  
RUN;  
  
PROC GCHART DATA = class;  
  PIE age / VALUE = ARROW  
  LEGEND DISCRETE;  
RUN;
```



ODS GRAPHICS FROM THE PIE CHART TASK

2D pie charts are available in PROC TEMPLATE, but not in the SG procedures, from SAS 9.3. However, the Pie Chart task in SAS Studio does not appear to include options to add a legend. Although the generated code can be copied and edited later.



Category variable: Age

```
/*--Define Pie Template--*/  
proc template ;  
  define statgraph WebOne.Pie;  
    begingraph;  
    layout region;  
      piechart category=Age /  
        start=0 centerFirstSlice=1  
        datalabellocation=Outside;  
    endlayout;  
  endgraph;  
end;  
run;  
  
/*--SGRENDER proc statement--*/  
proc sgrender template=WebOne.Pie  
  data=SASHELP.CLASS;  
run;
```

CONCLUSION

Before SAS Studio became available in SAS 9.4 it was only possible to generate PROC TEMPLATE code to create ODS Graphics output from the ODS Graphics Designer, or by saving the template code saved by PROC SGPLOT or PROC SGSCATTER. Now PROC SGPLOT code can be generated directly using SAS Studio tasks, so there really is little excuse for not trying out the generation of plots using ODS Graphics.

RECOMMENDED READING

Holland, Philip R. 2015. *SAS Programming and Data Visualization Techniques*. 1st ed. New York, NY, USA: Apress.com.

Holland, Philip R. 2014. "Something for Nothing! Converting Plots from SAS/GRAPH® to ODS Graphics". SAS Global Forum, P1610-2014. Washington, DC, USA.

Holland, Philip R. 2011. "Why Should You Be Using SG (Statistical Graphics) Procedures in SAS® 9.2?". SAS Global Forum, P427-2011. Las Vegas, NV, USA.

Holland, Philip R. 2009. "Using the ODS Graphics Designer to Create Your Own Templates in SAS® 9.2". PhUSE, TS05-2009. Basel, Switzerland.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name:	Philip R Holland
Organization:	Holland Numerics Limited
Address:	94 Green Drift
City, State ZIP:	Royston, Hertfordshire, SG8 5BT, United Kingdom
Work Phone:	+44-7714-279085
Fax:	+44-1763-242486
Email:	phil@hollandnumerics.com
Web:	www.hollandnumerics.com
sasCommunity.org :	www.sascommunity.org/wiki/User:Prholland

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.